

EBNF Grammar สำหรับภาษา CSubset

Factor = Designator | Number | "(" Expression ")" .
Term = Factor { ("*" | "/" | "%") Factor } .
SimpleExpr = ["+" | "-"] Term { ("+" | "-") Term } .
EqualityExpr = SimpleExpr [("<" | "<=" | ">" | ">=") SimpleExpr] .
Expression = EqualityExpr [("==" | "!=") EqualityExpr] .
ConstExpression = Expression .

FieldList = VariableDeclaration { VariableDeclaration } .
StructType = "struct" Ident ["{" FieldList "}"] .
Type = Ident | StructType .
IdentArray = Ident { " [" ConstExpression "] " } .
IdentList = IdentArray { " , " IdentArray } .
VariableDeclaration = Type IdentList " ; " .
ConstantDeclaration = " const " Type Ident " = " ConstExpression " ; " .

Designator = Ident { ("." Ident) | (" [" Expression "] ") } .
Assignment = Designator " = " Expression " ; " .
ExpList = Expression { " , " Expression } .
ProcedureCall = Ident "(" [ExpList] ") " " ; " .
IfStatement = "if" "(" Expression ") " "{" StatementSequence "}" ["else" "{"
StatementSequence "}"] .
WhileStatement = "while" "(" Expression ") " "{" StatementSequence "}" .
Statement = [Assignment | ProcedureCall | IfStatement | WhileStatement] .
StatementSequence = { Statement } .

FPSection = Type IdentArray .
FormalParameters = FPSection { " , " FPSection } .
ProcedureHeading = Ident "(" [FormalParameters] ") " .
ProcedureBody = { ConstDeclaration | VariableDeclaration } StatementSequence .
ProcedureDeclaration = "void" ProcedureHeading "{" ProcedureBody "}" .

Program = { ConstantDeclaration | VariableDeclaration } ProcedureDeclaration
{ ProcedureDeclaration } .

Number and Ident are terminal symbols. Program is the start symbol.